
دوال البحث المتقدمة .6

VLOOKUPدالة 

من أهم دوال البحث الموجودة في الاكسيل حيث تتميز بالسهولة والدقة. وهي  VLOOKUPدالة 
تبحث عن قيمة معينة في نطاق البحث بدلالة قيمة موجودة في العمود الأول من نطاق البحث. فلو 
كان لديك نطاق من البيانات خاص بالموظفين بحيث يظهر فيه رقم الموظف واسمه والقسم الذي 

البحث مثلاً عن اسم الموظف أو القسم الذي يعمل به بدلالة  فبالإمكانالبيانات،  يعمل الى آخره من
والشكل العام لدالة   رقمه الوظيفي بشرط أن يكون الرقم الوظيفي في العمود الأول من نطاق البحث.

VLOOKUP :هو كالتالي  

VLOOKUP(lookup_value; table_array; col_index_num; [range_lookup]) 

  حيث أنها تأخذ الوسائط/المعاملات التالية:

 Lookup_value القيمة التي سيتم البحث عنها في العمود بحث بدلالتهاالالمراد قيمة ال .
الأول من الجدول أو النطاق. 

   Table_arrayنطاق البحث
  col_index_num  رقم العمود المحتوي على نتيجة البحث 
  range_lookupم أو تقريبي. اذا كنا نبحث عن تطابق تام فيجب ادخال تحدد هل التطابق تا

. TRUE  أو 1واذا كنا نبحث عن تطابق تقريبي ندخل القيمة  FALSEأو    0هذا المعامل ك
.1في حال لم يتم ادخال هذا المعامل فان القيمة الافتراضية هي 

  ملاحظات/

 ذا كانت قيمة اrange_lookup  تساوي TRUE فيجب وضع قيم أو إذا كانت محذوفة ،
 VLOOKUPبترتيب فرز تصاعدي؛ وإلا فقد لا ترُجع  table_array العمود الأول لـ 
 .القيمة الصحيحة

   إذا كانت قيمة range_lookup تساوي   FALSE فلا داعي لفرز القيم في العمود الأول ،
 .table_arrayمن 

   إذا كانت قيمة الوسيطةrange_lookup   تساوي  FALSE ستعثر الدالة ،VLOOKUP  
  table_arrayعلى تطابق تام فقط. وفي حالة تطابق قيمتين أو أكثر في العمود الأول من 

، فسيتم استخدام أول قيمة يتم العثور عليها. إذا لم يتم العثور على lookup_valueمع 
  N/A# تطابق تام، يتم إرجاع قيمة الخطأ

  قان الدالة ترجع أكبر قيمة أقل من أو تساوي القيمة المراد في حال استخدام التطابق التقريبي
البحث بدلالتها.

مع التطابق التام VLOOKUPمثال على استخدام دالة 

 B2نطاق البيانات الظاهر يبين لنا رقم الموظف ورقمه والمطلوب هو إيجاد اسم الموظف في الخلية 
  للوصول لهذا الهدف نستخدم المعادلة التالية:. B1بدلالة رقمه الذي سيتم إدخاله في الخلية 

=VLOOKUP(B1;D1:E11;2;FALSE) 

الذي تم إدخاله في  –تحتوي على لقيمة المراد البحث بدلالتها (رقم الموظف  B1حيث أن الخلية 
هو رقم  2يمثل نطاق البحث، والمعامل الثالث  D1:E11العمود الأول من نطاق البيانات/البحث) و 

 FALSEحتوي على نتيجة البحث (العمود الخاص باسم الموظف) والوسيط الأخير قيمته العمود الم
  حيث أننا نبحث عن تطابق تام. 0أو 

  مع التطابق التقريبي VLOOKUPمثال على استخدام 

موظفي احدى الشركات والمطلوب هو إيجاد نسبة الضريبة نطاق البيانات الظاهر يوضح بيانات 
لكل موظف. في هذه الحالة لا يمكننا استخدام التطابق التام حيث لا يوجد مبالغ محددة للرواتب 

  وبالتالي نلجأ للتطابق التقريبي.

لالتها)   دأولاً تم اعداد نطاق البحث بالشكل الملائم حيث وضعنا  قيم الراتب (القيمة المراد البحث ب 
ا العمود تصاعدياً. في هذه الحالة فإن  العمود الأول هذفي العمود الأول من نطاق البحث وتم فرز 

سيطبق عليه نسبة  5000من نطاق البحث يمثل الشرائح الضريبية فمثلاً الموظف ذو الراتب أقل من 
  وهكذا. %5سيطبق عليه نسبة الضريبة  2500  حتى اقل من   5000و من  %1الضريبة 

  المعادلة المطلوبة هي كالتالي:

=VLOOKUP(J2;$L$1:$M$10;2;1) 

عن أكبر قيمة أقل من أو تساوي القيمة  L1:M10وفيها يتم البحث في العمود الثاني في النطاق 
في المعادلة حتى يتسنى  L$1:$M$10$  الراتب)، تم تثبيت خلايا النطاق ( J2الموجودة في الخلية 

 باقي الخلايا. لنا تعميمها على

1-6الشكل  

  مع أكثر من نطاق للبحث   VLOOKUPاستخدام دالة 

مع أكثر من نطاق بحث لجعل الدالة أكثر مرونة وفعالية   VLOOKUPمن الممكن استخدام دالة 

البيع لكل مندوب من مندوبي المبيعات اعتماداً على عدد المطلوب في هذا المثال إيجاد نسبة عمولة 
سنوات خدمته حيث أنه هناك جدولين للبحث أحدهما خاص بمندوبي المبيعات ذوي عدد سنوات خدمة 

 3والآخر خاص بذوي سنوات خدمة أكبر من أو يساوي  Table1وقد تم تسميته  سنين 3أقل من 
  لوبة هي كالتالي:. المعادلة المطTable2سنوات وقد تسميته 

=VLOOKUP(C2;IF(B2<3;Table1;Table2);2) 

2-6الشكل  

يتم على أساسها تحديد  IFفي هذه المعادلة تم استبدال الوسيط الثاني للدالة (نطاق البحث) بمعادلة 
ومن ثم تعميمها على باقي خلايا حقل  D2نطاق البحث المطلوب. يتم ادخال هذه المعادلة في الخلية 

  نسبة العمولة.

MATCHدالة 

ترجع الموقع النسبي لقيمة ما ضمن نطاق/سلسلة من القيم. على سبيل المثال لدينا   MATCHدالة 
يشكل مجموعة من أرقام الهويات  والمطلوب  Aمن البيانات  في العمود  الظاهر نطاقفي الشكل 

لهذا الغرض دالة  النطاق. نستخدمضمن هذا  C2معرفة الموقع النسبي للقيمة الموجودة في الخلية 
Match :التالية

=MATCH(C2;A2:A31;0) 

ق االثاني عبارة عن نط والمعامل حيث أن المعامل الأول للدالة هو القيمة المراد البحث عنها/بدلالتها 
 0أو  1البحث/البيانات و المعامل الثالث يحدد هل التطابق المطلوب تام أو تقريبي وهو يأخذ القيمة 

  1-أو 

 3-6الشكل 

  ملاحظة:

، فان الدالة تبحث عن أكبر قيمة تكون أصغر من أو تساوي القيمة 1كانت قيمة الوسيط الثالث هي  اذا
التي يتم البحث عنها (المعامل الأول) وفي هذه الحالة يجب ترتيب النطاق بشكل تصاعدي. واذا 

 م البحثتبحث عن أول قيمة تكون مطابقة تماما للقيمة التي يت  Matchفان الدالة  0كانت القيمة 
تبحث عن أصغر قيمة تكون أكبر من أو تساوي   MATCHفان الدالة  ‐1عنها. واذا كانت القيمة 

القيمة  التي يتم البحث عنها وفي هذه الحالة يجب ترتيب النطاق بشكل تنازلي. اذا تجاهلت الوسيط 
  .1الثالث فان الدالة تفترضه 

4-6 الشكل 

  هل القيمة المحددة موجودة ضمن نطاق محدد أم لا لتحديد  Matchاستخدام دالة 

موجود  C2لمثال هو تعديل للمثال السابق و المطلوب هو تحديد هل الرقم الموجود في الخلية هذا ا
  Aضمن الأرقام الموجودة في العمود 

م وبالتالي سوف نستخد NA#ترجع قيمة الخطأ  فإنهااذا لم تجد تطابق   MATCHهي أن الدالة الفكرة 
هو خطأ أم لا. في حال كانت النتيجة خطأ هل  MATCHالتي سوف تفحص هل ناتج الدالة   IFدالة 

  فان الرقم "غير موجود" والا فانه موجود وعلى ذلك نكتب المعادلة التالية:

=IF(ISERROR(MATCH(C2;A:A;0));" موجود"؛"موجود غير ") 

  خطأ أم لا.  MATCHلفحص هل نتيجة الدالة  ISERROR حيث استخدمنا دالة 

INDEXدالة 

كنا نريد القيمة  إذافمثلا   ترجع القيمة الموجودة في موقع نسبي معين في نطاق محدد. INDEXدالة 
  نكتب الدالة التالية:) 5-6(شكل  C3:C7التي تقع في الموقع الرابع من النطاق 

=INDEX(C3:C7;4) 

ل الأول عبارة عن نطاق البيانات التي تقع القيمة تأخذ ثلاث معاملات. المعام INDEXحيث أن الدالة 
المراد ارجاعها بداخله، و المعامل الثاني هو رقم الصف  للقيمة المرادة و المعامل الثالث وهو معامل 
اختياري عبارة عن رقم العمود للقيمة المراد ارجاعها وبما أن نطاق البيانات في مثالنا عبارة عن 

استخدام المعامل الثالث. أما لو أردنا الحصول على القيمة الموجودة في عمود واحد فقط فلا يلزم 
  فنكتب المعادلة التالية: C3:J7من النطاق  والعمود الرابعلصف الثالث 

=INDEX(C3:J7;3;4) 

  حيث أننا استخدمنا المعامل الثالث في هذه الحالة.

لاستخراج قيمة معينة من نطاق ما  INDEXمثال: استخدام دالة 

يوضح تكلفة الشحن بناء على منطقة  C3:J7نطاق البيانات  )6-6(شكل  في ورقة العمل الظاهرة
الشحن وحجم العنصر المراد شحنه. والمطلوب هو الحصول عل تكلفة شحن العناصر الموجودة في 

  يظهر حجم العنصر و منطقة الشحن الخاصة به. Rو  Qحيث انه في العمودين    Pالعمود 

  للوصول للهدف نكتب المعادلة التالية:

=INDEX($C$3:$J$7;Q2;R2) 

الثاني عبارة عن رقم  حيث المعامل الأول هو نطاق البحث المحتوي على أسعار الشحن و المعامل
  الصف المحتوي على النتيجة والثالث رقم العمود. ثم نعمم الصيغة على باقي الخلايا.

5-6 الشكل

  عمليات بحث فعالة و مرنة لإنشاء INDEX‐MATCHاستخدام دالتي 

بشكل منفصل ورأينا استخداما مفيدا لكل منهما  MATCHو  INDEXتعرفنا على كيفية استخدام دالتي 
  ولكن قوة هاتين الدالتين الحقيقية تكمن في استخدامهما مع بعضهما البعض.

 INDEX‐MATCHمثال على اجراء البحث باستخدام 

لدينا نطاق من البيانات يتكون من حقلين الحقل الأول خاص برقم الضمان الاجتماعي  7-6في الشكل 
 P2و الثاني خاص بالاسم والمطلوب هو اظهار اسم الشخص  ذو رقم الضمان الموجود في الخلية 

  .  Q2وذلك في الخلية َ

و رقام الضمان وهوللوصول لهذه النتيجة يجب أن نحدد أولا الموقع النسبي لرقم الضمان في عمد ا
 التالية: MATCHوذلك من خلال دالة   Mالعمود 

=MATCH(P2;M:M;0) 

للحصول على الاسم الموجود في الموقع الذي حصلنا علية من خلال دالة  INDEXثم نستخدم دالة 
MATCH  وذلك من قائمة الأسماء الموجودة في العمودN.   :من خلال المعادلة التالية 

=INDEX(N:N;MATCH(P2;M:M;0)) 

6-6 الشكل

  :ملاحظة

‐INDEXولكن استخدام دالتي  VLOOKUPيمكن الحصول على نفس النتيجة السابقة من خلال دالة 

MATCH  أكثر مرونة حيث ان دالةVLOOKUP  أن القيمة المراد البحث بدلالتها يجب أن  تشترط
  . INDEX‐MATCHتكون في العمود الأول من نطاق البحث بينما هذا القيد غير موجود  في 

  INDEX‐MATCHتنفيذ البحث ثنائي الاتجاه من خلال دالتي 

 نفي هذا المثال لدينا نطاق من البيانات يوضح لنا مبيعات مجموعة من المنتجات في مجموعة م
الأشهر والمطلوب هو إيجاد مبيعات منتج محدد لشهر محدد. ونحن قد سمحنا للمستخدم بإدخال المنتج 

  . B8والمطلوب إيجاد المبيعات في الخلية   B2و   B1والشهر المطلوبين في الخلايا 

7-6 الشكل 

النتيجة يجب ان نحدد أولا رقم الصف الذي يقع فيه الشهر المطلوب (انزياح وللوصول الى هذه 
  الشهر) ورقم لعمود الذي يقع به المنتج (انزياح المنتج).

  :B4في الخلية   التالية MATCHللحصول على ازياح الشهر نستخدم دالة 

=MATCH(B1;E2:E14;0) 

  يمثل نطاق الأشهر. E2:E12تمثل الشهر المطلوب و  B1حيث أن 

  :B5في الخلية   MATCHللحصول على انزياح المنتج نستخدم دالة 

=MATCH(B2;F1:J1;0) 

  يمثل نطاق المنتجات. F1:J1و    تحتوي على المنتج المطلوب  B2حيث أن 

 B5و   B4الان بعد الحصول على العمود و الصف اللذان تقع النتيجة عند تقاطعهما في الخلايا 
للشهر في  B1للحصول على مبيعات المنتج الموجود في الخلية   B6في الخلية  INDEXنستخدم دالة 

B2 :كالتالي  

=INDEX(F2:J14;B4;B5) 

8-6 الشكل

هو نطاق البحث . F2:J14حيث 

  :B8وللحصول على النتيجة من خلال معادلة واحدة نكتب المعادلة التالية في الخلية 

=INDEX(F2:J14;MATCH(B1;E2:E14;0);MATCH(B2;F1:J1;0)) 

INDEX‐MATCHدالتي  الأعمدة باستخداماجراء البحث ثنائي 

عملية البحث بناء على قيم موجودة في أكثر من حقل في نطاق  لإجراءأحيانا قد تكون مضطراً 
نطاق البيانات ) 11-6(شكل البحث في نفس اللحظة. فعلى سبيل المثال في ورقة العمل الظاهرة 

D1:F12  يوضح لنا مجموعة من أنواع السيارات و موديلاتها و الكود الخاص بكل سيارة و المطلوب
ففي هذه الحالة  B1:B2هو الحصول على كود السيارة التي نوعها و موديلها مدخلين في الخليتين 

  ة.دالمطلوب اجراء عملية البحث بناء على قيمتين في حقلين مختلفين  و ليس قيمة واحدة كما هي العا

9-6 الشكل

  يمكن اجراء هذه العملية بأكثر من طريقة

  هي ادراج حقل جديد في نطاق البيانات محتوياته هي عبارة عن  الطريقة الأولى    

ومن ثم سوف نستخدم هذا الحقل   12-6محتويات حقل النوع والموديل كما هو ظاهر في الشكل 
  للبحث على أساسه. حسب المعادلة التالية:

=INDEX(G1:G12;MATCH(B1&B2;F1:F12;0)) 

11-6 الشكل
10‐6 الشكل

12-6 الشكل 

لتحديد موقع الكود المطلوب النسبي ضمن قائمة الأكواد  MATCHفي هذه المعادلة استخدمنا دالة 
  .MATCHلاستخراج هذا الكود بناء على موقعه الذي حصلنا عليه من  INDEXومن ثم دالة 

في هذه الدالة المعامل  MATCH(B1&B2;F1:F12;0)  الداخلية MATCHلنتوقف قليلا عند دالة 
قيمة المراد البحث بدلالتها) عبارة عن النوع والموديل معا (قيمتين من حقلين مختلفين) تم الأول (ال

 حقليهو عبارة عن حاصل ربط  F1:F12. ونطاق البحث &ربطهما معا باستخدام معامل الارتباط 
  النوع والموديل كما رأينا سابقً.

لاجراء البحث ثنائي الأعمدة هي باستخدام صيغ المصفوفات وفيها سوف نستغني  الطريقة الثانية
الذي أضفناه في الطريقة الأولى (النوع والموديل) عن طريق انشاء هذا الحقل  الحقل المرحليعن 

  في الذاكرة. وذلك حسب معادلة لمصفوفات التالية:

=INDEX(F1:F12;MATCH(B1&B2;D1:D12&E1:E12;0)) 

  هذه المعادلة. لإدخال    Ctrl+Shift+Enterولا تنسى أن تضغط 

13-6 الشكل

حصلنا عليه من  MATCHهذه المعادلة هي نفس المعادلة السابقة ولكن نطاق البحث بالنسبة لدالة 
 خدمهاوالتي سوف ينتج عنها مصفوفة في الذاكرة سوف تست D1:D12&E1:E12خلال المعادلة  

  كنطاق للبحث.  MATCHدالة 

مع أكثر من نطاق للبيانات  INDEXاستخدام دالة 

الحالة الأولى تتعامل مع نطاق بيانات واحد وهي الحالة  15-6لها حالتين (كما بالشكل   INDEXدالة 
التي تعاملنا معها في الأمثلة السابقة والحالة الثانية تتعامل مع أكثر من نطاق في نفس الوقت وتأخذ 

  البنية و المعاملات/الوسائط التالية:

INDEX(reference, row_num, [column_num], [area_num]) 

14-6 الشكل

15‐6 الشكل

o b e i k a n d l . c o m



  دوال البحث والمراجع المتقدمة  الفصل السادس

83 

:على الوسيطات التالية INDEX يحتوي بناء جملة الدالة

 Reference     وسيطة مطلوبة. مرجع إلى نطاق خلايا واحد أو أكثر.
   Row_num    وسيطة مطلوبة. رقم الصف في المرجع الذي سيتم إرجاع مرجع منه. 
  Column_num     اختيارية. رقم العمود في المرجع الذي سيتم إرجاع مرجع منهوسيطة.
   Area_num     وسيطة اختيارية. تحديد نطاق في مرجع سيتم إرجاع تقاطع Row_num 

، والثانية 1يتم ترقيم الناحية الأولى التي حددتها أو أدخلتها بالرقم  .منه  Column_num و
.1الناحية  INDEX فتستخدم الدالة، Area_num ، وهكذا. إذا تم حذف2بالرقم 

 ، فتكون(A1:B4,D1:E4,G1:H4)الخلايا  Reference على سبيل المثال، إذا وصفت الوسيطة
Area_num 1 النطاق A1:B4وتكون ، Area_num 2 النطاق D1:E4 وتكون Area_num3  

 G1:H4 النطاق

  مع أكثر من نطاق INDEXمثال على استخدام 

توضح لنا مبيعات مجموعة من المحاصيل لأشهر معينة لعدة  16-6 في الشكلورقة العمل الظاهرة 
محدد ومنطقة محددة. فبهذا يكون المطلوب  مناطق والمطلوب إيجاد مبيعات محصول محدد لشهر

متعددة النطاقات كالتالي (مع   INDEXبحث ثنائي الاتجاه متعدد النطاقات. نستخدم لهذا الغرض دالة 
  ) : banana, pear, and appleنا نطاقات المحاصيل كالتالي:العلم بأننا أسمي

=INDEX((banana;pear;apple);MATCH(N5;A1:A6;0);MATCH(N4;A1:J1;0);IF(N

3="banana";1;IF(N3="pear";2;3))) 

، الوسيط الثاني و الثالث INDEX/معامل أول للدالة البحث كوسيطحيث أننا أدخلنا أسماء نطاقات 
كما هو ظاهر في المعادلة، أما الوسيط  MATCHلعمود) حصلنا عليهم من خلال دالة (رقم الصف و ا

والتي  IFالأخير للدالة و الذي من خلاله نحدد للدالة نطاق البحث المطلوب فحددناه من خلال معادلة 
 3 القيمةو  Pearلمحصول  2و القيمة  Bananaاذا كان المحصول  1سوف ترجع لنا القيمة 

. Appleلمحصول 

و  3سوف ترجع الرقم  IFفان دالة   Appleهو  N3فمثلا اذا كان المحصول المدخل في الخلية 
  . Appleبالتالي فنطاق البحث المطلوب البحث بداخله هو النطاق الثالث 

INDIRECTدالة 

الدوال القوية والمرنة التي نستطيع من خلالها انشاء معادلات قوية ومرنة .  من  INDIRECTدالة 
 وهي تأخذ نص كوسيط وترجع المرجع الذي له نفس الاسم.

  فمثلا الدالة التالية:

=INDIRECT(“a2”) 

وبالتالي سوف تظهر  A2له نفس الاسم وهو  المرجع الذيتقيم النص المدخل لها (الوسيط) فترجع 
.  INDIRCTفي الخلية التي كتبنا بها دالة  A2خلية محتويات ال

16-6 الشكل

في ورقة العمل الظاهرة في الشكل فسترجع لنا ما في الخلية    J2فلو طبقنا الدالة السابقة في الخلية 
A2.  

الذي نريد استخدامه   Reference textعلى النص  للحصول  &ومن الممكن استخدام معامل الربط 
  INDIRECTكوسيط للدالة 

  كمثال على ذلك:

=INDIRECT(J4&J3)  

 B3فتكون النتيجة هو محتوى الخلية 

17-6 الشكل 

18‐6 الشكل

o b e i k a n d l . c o m



  دوال البحث والمراجع المتقدمة  الفصل السادس

86 

  من عدة أوراق عمل بشكل مرن تلخيصيهعرض بيانات  :مثال 

لدينا مجموعة من أوراق العمل بحيث تحتوي كل ورقة عمل  19-6 المصنف الظاهر في الشكلفي 
مجموع المبيعات  علىتحتوي  INDIRECT‐1العمل المسماة  معينة وورقةالمبيعات الخاصة بمنطقة 

  للمناطق الأربعة.

بحيث  INDIRECT‐1والمطلوب هو أن نعرض مجموع مبيعات كل منطقة في صفحة الملخص 
  .B10:B12ثم نعمم على الخلايا  B9نكتب المعادلة الأولي في الخلية 

من ورقة العمل الخاصة بهذه  G6الخلية التي تحتوي على مجموع مبيعات منطقة معينة هي الخلية 
من الممكن الحصول عليه من خلال المعادلة  Eastالمنطقة فمثلا مجموع مبيعات المنطقة الشرقية 

  التالية:

=East!G6 

19-6 الشكل

  وبالتالي بالإمكان كتابة المعادلة التالية للحصول على مجموع مبيعات للمنطقة الشرقية:

=INDIRECT(A9&"!"&"g6") 

أو  50ومن ثم تعميمها على باقي الخلايا للحصول على النتيجة المطلوبة. لاحظ هنا انه لو كان لدينا 
منطقة فسيكون المطلوب هو فقط التعميم على باقي الخلايا وهذا بسبب المرونة العالية لدالة  100

INDIRECT.  

20-6 الشكل 

  قائمة منسدلة معتمدة على قائمة أخرى لإنشاء  INDIRECTدالة  استخدام :2مثال 

في ورقة العمل الظاهرة لدينا نطاق ممن البيانات يوضح مجموعة من أسماء الولايات وأسماء المدن 
ثم انشاء قائمة منسدلة  C2التابعة لكل ولاية والمطلوب هو انشاء قائمة بأسماء الولايات في الخلية 

  .C2تظهر أسماء المدن الخاصة بالولاية الظاهرة في الخلية  D2في الخلية 

لهذا الهدف نسمي أولا ً نطاقات الخلايا المحتوية على أسماء المدن لكل ولاية بحيث مثلاً نطاق 
. وأسهل طريقة لتسمية AZب  G3:Q3الخلايا الذي يحتوي على أسماء المدن الخاصة بولاية أريزونا 

 ثم الذهاب الى شريط  F3:Q14الخلايا هي تحديد نطاق الولايات والمدن  Formulas ى ثم ال
Defined names  (الأسماء المعرفة)   ثمCreate From Selection  (الانشاء من التحديد)

 Left columnفيظهر مربع الحوار الخاص بإنشاء أسماء من التحديد نحدد الخيار الخاص ب 
  (العمود الايسر) حتى يأخذ كل صف الاسم الموجود في  الخلية الموجودة في أقصى يساره.

21-6 الشكل 

ثم نذهب الى شريط  B3تسرد أسماء الولايات نقف في الخلية   B3ولانشاء قائمة منسدلة في الخلية 
Data  (بيانات) ثمData Validation  (التخقق من البيانات) من قائمة Allow    (السماح) نختار
List  (قائمة) ونحدد عناصر القائمة بالنطاقF3:F14.  

 C3نقف بداخل الخلية  B3معتمدة على القيمة الظاهرة في  C3الآن لانشاء قائمة منسدلة في الخلية 
بع النص الخاص ب وفي مر  Allowمن قائمة   Listنختار   Data validationثم نذهب الى 

source :نكتب المعادلة التالية  

22-6 الشكل

23-6 الشكل 

=INDIRECT(B3) 

. B3هنا ترجع المرجع الذي له نفس الاسم الموجود في الخلية   INDIRECTحيث أن دالة 

وللتخلص من الفراغات التي قد تظهر في بعض قوائم المدن، نحدد نطاق الولايات والمدن ثم نذهب 
(الذهاب الى خاص) ثم   Go to specialحث وتحديد) ثم (ب Find and selectثم  Homeالى 

  . Delete الماوس الأيمن ونختار زرب(الفراغات). نضغط موافق ثم نضغط  Blanksنختار 

24‐6 الشكل

  لتنفيذ البحث ثنائي الاتجاه  INDIRECTاستخدام دالة 

وفي هذا المثال  INDEX‐MATCHسابقاً نحن أجرينا عملية البحث ثنائي الانجاه من خلال دالتي 
  INDIRECTسنتعرف على كيفية الحصول على نفس النتيجة بصورة أسهل من خلال استخدام دالة 

  المثال:

 نطاق البيانات الظاهر يوضح مبيعات مجموعة من المنتجات لمجموعة من الأشهر و المطلوب هو
  ارجاع مبيعات منتج معين لشهر محدد.

الفكرة هي أنه بإمكاننا الحصول على النتيجة المطلوبة من خلال إيجاد تقاطع الصف المحتوي على 
  الشهر المطلوب مع العمود المحتوي على المنتج المطلوب.

بهذه العملية نسمي الصفوف والأعمدة بحيث أن يأخذ كل صف اسم الشهر الخاص به وكل وللقيام 
  عمود اسم المنتج الخاص به. وأسهل طريقة لتسمية الصفوف والأعمدة هي كالتالي:

  بالنسبة للصفوف نحدد النطاقD2:H14  ثم نذهب الى تبويبFormulas  ثم الى الأسماء
 Create Fromنضغط على انشاء من التحديد  Definbed Namesالمعرفة 

Selection  ثم نحدد على مربع الاختيارLeft Column. 

25-6 الشكل 

  بالنسبة للأعمدة نحدد النطاقE1:H14  ثم نذهب الىFormulas  ثمCreate From 

Selection   ثم نحدد مربع الاختيارTop Row.

  ثم نستخدم المعادلة التالية للحصول على النتيجة:

=INDIRECT(B1) INDIRECT(B2) 

إلى ما يقابله من  B2و  B1لتحويل ما يوجد في الخلايا  INDIRECTفي هذه المعادلة استخدمنا دالة 
نطاق ثم استخدمنا معامل التقاطع وهو المسافة (" ") للحصول على تقاطع صف الشهر المحدد 

  ).B2)  مع عمود المنتج المطلوب (الموجود في B1(الموجود في 

OFFSETدالة 

هي أحد دوال المراجع وهي ترجع مرجع لنطاق يتكون من عدد محدد من الصفوف  OFFSETدالة 
و الأعمدة  ويبعد عدد معين من الصفوف و الأعمدة عن خلية محددة أو نطاق من الخلايا. وتأخذ 

  البنية التالية:

OFFSET(reference, rows, cols, [height], [width]) 

:التالية الوسائط/المعاملاتعلى  OFFSET جملة الدالةيحتوي بناء 

 Reference      مطلوبة. المرجع الذي تريد أن تستند منه الإزاحة. يجب أن يشير المرجع
 .#VALUE! قيمة الخطأ OFFSET إلى خلية ؛ بخلاف ذلك، ترُجع الدالة

 Rows      .ة ه/ارجاعه عن الخليعدد الصفوف التي يبعد عنها النطاق المطلوب تحديدمطلوبة
المرجع. اذا كان الرقم بالموجب فهذا يعني ان التحرك يتم للأسفل واذا كان بالسالب فالتحرك 

 للأعلى.

    Cols    عدد الأعمدة التي يبعد عنها النطاق المطلوب عن الخلية المرجع. اذا كان بالموجب
(في ورقة عمل اتجاهها من اليسار  لليسارواذا كان بالسالب فالتحرك  لليمينفان التحرك 

لليمين).
  Height     الذي يتم إرجاعه للنطاقاختيارية. الارتفاع، في عدد الصفوف، الذي تريده. 

     Width الذي يتم إرجاعه. يجب  للنطاق اختيارية. العرض، في عدد الأعمدة، الذي تريده

 26-6في الشكل  A1نا معرفة محتوى الخلية التي تبعد أربعة صفوف للأسفل من الخلية دفلو أر
  بالإمكان استخدام الدالة التالية:

=OFFSET(A1;4;0) 

هي الخلية المرجع، الوسيط الثاني هو عدد الصفوف الذي تبعد الخلية  A1حيث أن الوسيط الأول 
المطلوب ارجاعها بمقداره عن الخلية المرجع وبما انه الاتجاه للأسفل استخدمنا علامة الموجب، 

لأننا نريد التحرك على نفس العمود، الوسيطين الرابع و الخامس بالإمكان  0الوسيط الثالث هو 
ي هذه المثال حيث أن النطاق المطلوب ارجاعه يتكون من خلية واحدة و بالتالي الاستغناء عنهما ف

 لايوجد له طول وعرض.

26-6 الشكل 

 B1ويبعد عن الخلية  وعمود واحدصفوف  3نا معرفة مجموع النطاق المكون من دلو أرمثال آخر و
  نستخدم المعادلة التالية: العمودصفوف على نفس  4بمقدار 

=SUM(OFFSET(B1;4;0;3;1)) 

ثم أدخلنا هذا النطاق على دالة  OFFSETحيث أننا حصلنا أولاً على النطاق المطلوب من خلال دالة 
SUM للحصول على النتيجة المطلوبة.  

وذلك   Aحيث المطلوب هو ارجاع التاريخ الأخير في العمود  OFFSETعلى دالة  ولنأخذ مثالاً اخر
  . ونستخدم لذلك المعادلة التالية:D2في الخلية 

=OFFSET(A1;COUNTA(A:A)‐1;0) 

هو عدد غير ثابت وبالتالي  نتحركهكخلية مرجع، عدد الصفوف الذي نريد أن  A1حيث استخدمنا 
ا حساب عدد الخلايا غير يتم فيه والتي COUNTA(A:A)‐1  حصلنا عليه من خلال معادلة هي

27‐6 الشكل

للوسيط  0. واستخدمنا A1مطروحاً منه واحد لأننا نريد العد من بعد الخلية  Aي العمود الفارغة ف
 النطاق المُرجع والخامس لأنالثالث لأننا نتحرك على نفس العمود، وتم اهمال الوسيطين الرابع 

  عبارة عن خلية واحدة.

  لو أردنا أن نعرف متوسط سعر الاغلاق لأخر ثلاث أيام نستخدم المعادلة التالية:: مثال آخرو

=AVERAGE(OFFSET(B1;COUNTA(B:B)‐1;0;‐3;1)) 

  النطاق المطلوب إيجاد متوسط خلاياه كالتالي: لإرجاع   OFFSETحيث أننا استخدمنا دالة 

OFFSET(B1;COUNTA(B:B)‐1;0;‐3;1) 

كخلية مرجع، وحصلنا على عدد الأعمدة الذي نريد أن نتحركه بدءاً من خلية   B1وفيها تم استخدام  
لأننا نتحرك على نفس العمود،  0، الوسيط الثالث COUNTA(B:B)‐1 المرجع من خلال معادلة

(في حالتنا هي  Bوعلامته سالب لأننا تحركنا حتى الخلية الأخيرة في العمود  3الوسيط الرابع هو 
) وكي نحدد النطاق المطلوب نريد التحرك للأعلى ثلاث صفوف وأن يكون عدد أعمدة B9الخلية 

  (الوسيط الخامس). 1النطاق المُرجع 

28‐6 الشكل

CHOOSEدالة 

هي احدى دوال البحث والمراجع وهي ترجع قيمة من بين مجموعة من القيم بدلالة  CHOOSEدالة 
  موقع هذه القيمة بالنسبة لباقي القيم. وهي تأخذ البنية التالية:

CHOOSE(index_num; value1; [value2], ...) 

يمثل رقم القيمة المطلوب ارجاعها بالنسبة لمجموعة  index_numحيث أن الوسيط الأول للدالة 
  قيمة. 254أن تأخذ حتى  CHOOSEالقيم المعطاة. باقي الوسائط تمثل مجموعة القيم، تستطيع دالة 

  على سبيل المثال الدالة التالية:

=CHOOSE(3;B6;C6;D6;E6;F6;G6) 

  .119من بين القيم الموجودة وهي  3التي ترتيبها ترجع قيمة الخلية 

29-6 الشكل 

CHOOSEمثال على استخدام دالة 

يوضح لنا مجموع مبيعات محصول معين لمجموعة من  30-6نطاق البيانات الظاهر في شكل 
) K4الأشهر، والمطلوب هو إيجاد مجموع المبيعات حتى شهر محدد (سوف يتم إدخاله في الخلية 

  .K8وذلك في الخلية 

  وللحصول على هذه النتيجة نستخدم المعادلة التالية:

=SUM(B6:CHOOSE(K4;B6;C6;D6;E6;F6;G6)) 

والتي تحتوي على مجموع المبيعات لشهر  B6هذه المعادلة يمتد من الخلية حيث أن نطاق الجمع في 
، هذه الخلية حصلنا K4وحتى الخلية التي تحتوي على مجموع مبيعات الشهر المحدد في الخلية يناير 

حيث أن الوسيط CHOOSE(K4;B6;C6;D6;E6;F6;G6)    التالية CHOOSEعليها من خلال دالة 
) وباقي وسائط الدالة K4رقم القيمة المطلوب ارجاعها (وهي القيمة في الخلية  الأول لهذه الدالة هو

  هي عبارة عن الخلايا المحتوية على مبيعات كل شهر من الأشهر.

إلى قائمة منسدلة تحتوي  K2ومن الممكن جعل ورقة العمل أكثر فاعلية عن طريق تحويل الخلية 
  لتصبح كالتالي:على أسماء الأشهر ومن ثم تعديل معادلة الناتج  

=SUM(B6:CHOOSE(MATCH(K4;N3:N14;0);B6;C6;D6;E6;F6;G6)) 

30-6 الشكل 

بدلاً من استخدام رقم الشهر بشكل صريح استخدمنا اسم الشهر وهي نفس المعادلة السابقة ولكننا 
  للحصول على رقم الشهر. MATCHدالة  استخدمناومن ثم 

ملاحظات: 

  للمزيد حول دالةMATCH .يرجى مراجعة الفقرة الخاصة بهذه الدالة في هذال الفصل  
 نطاقالأدخلنا أسماء الأشهر في  قائمة منسدلة بالأشهر لإنشاء N3:N14  ثم استخدمنا ميزة

للحصول على قائمة منسدلة بناء على القيم في   Data Validationالتحقق من البيانات 
ثم  Data Validationثم   بيانات Dataمن خلال الذهاب إلى تبويب خلايا نطاق الأشهر 

وندخل  Allowفي خانة  Listنحدد   Data Validationمن مربع الحوار الخاص ب 
)32-6(شكل  Sourceالنطاق المحتوي على أسماء الأشهر في خانة 

31‐6 الشكل

32o b e i k a n d l . c o m-6 الشكل 




